Supporting material

Maxwell's Mixing Equation Revisited: Characteristic Impedance Equations for Ellipsoidal Cells

Marco Stubbe[1] and Jan Gimsa[1,*]

[1]Chair of Biophysics, University of Rostock, Rostock, Germany

e-mail: marco.stubbe@uni-rostock.de, Phone: +49-381-498 6027;

e-mail: jan.gimsa@uni-rostock.de, Phone: +49-381-498 6020

Maple 12© code

derivation of simplified impedance equations for suspensions of ellipsoidal objects based on the model of Gimsa and Wachner (27,28)

Code developed by Dr. Marco Stubbe
An RC-model was introduced into Maxwell-Wagners mixing equation based on the axial longitudinal element (ALE) model for ellipsoidal single shell models

variables

Zi, Ze and Zm are the impedances of the internal and external media as well as the cell membrane
the indices a, b and c refer to the three principal axes with their lengths of ra, rb and rc
xia, xib and xic are the relative influential radii along the semi axes
xi is the relative influential radius of the larger sphere
omega is the circular frequency
A is the cross sectional area of each of the volume elements of the internal, membrane and external media. A is arbitrarily small and equal and constant along each principal axis
Cm and gm are the sectoral capacity and the area-specific membrane conductance
p is the volume fraction of the objects

clearing internal memory
```
> restart;
```

general simplifications

```
> gm:=0:
```

defining impedances

defining the impedances for all media elements
```
> Zia:=ra/(A*(si+I*omega*ei)):
Zea:=xia*ra/(A*(se+I*omega*ee)): Zm:=1/(A*(gm+I*omega*Cm)):
> Zib:=rb/(A*(si+I*omega*ei)):
Zeb:=xib*rb/(A*(se+I*omega*ee)):
```

```
> Zic:=rc/(A*(si+I*omega*ei)):
Zec:=xic*rc/(A*(se+I*omega*ee)):
```

Clausius-Mossotti-factor

defining the Clausius-Mossotti-factors (CMF) for the semi axes a, b, and c

simplifications:
the CMF can be simplified for different characteristic frequencies, depending on the electrical simplifications applied to the complete equivalent circuit (28)
e.g. Zm can be neglectet for frequencies beyond membrane dispersion
HINT: removing impedance components from the CMF (e.g. by converting these impedance components to text) is a very fast and simple method to consider the model properties in certain frequency bands

```
> CMFa:=(1+xia)/xia*(1-(Zia+Zm)/(Zia+Zm+Zea)*(1+xia)):
CMFb:=(1+xib)/xib*(1-(Zib+Zm)/(Zib+Zm+Zeb)*(1+xib)):
CMFc:=(1+xic)/xic*(1-(Zic+Zm)/(Zic+Zm+Zec)*(1+xic)):
> CMFa:=simplify(CMFa,symbolic):
CMFb:=simplify(CMFb,symbolic): CMFc:=simplify(CMFc,symbolic):
```

derivation of the CMF for randomly oriented objects from the three semi axis components

```
> CMF:=simplify((CMFa+CMFb+CMFc)/3,symbolic):
```

combing Maxwells mixing equation with the ALE model

simplifications:

```
> xi:=1/2:  # the relative influential radius of the large sphere is 1/2
> s:=solve((s-(se+I*omega*ee))/(xi/(1+xi)*s+1/(1+xi)*
(se+I*omega*ee))=p*CMF,s):
```

expressions for the conductivity and permittivity of the suspension for the DC case

```
> sigma_low:=collect(simplify(limit(evalc(Re(s)),omega=0),
symbolic),[p]);
> epsilon_low:=collect(simplify(limit(evalc(Im(s))/omega,
omega=0),symbolic),[r,Cm,p]);
```

expressions for the conductivity and permittivity of the suspension for infinitely high frequencies

reinitializing variables

```
> p:='p': xi:='xi': omega:='omega': gm:='gm': ee:='ee':
ei:='ei':
```

calculation of conductivity and permittivity

```
> sigma_inf:=collect(simplify(limit(evalc(Re(s)),omega
=infinity),symbolic),[p,xi]);
> epsilon_inf:=collect(simplify(limit(evalc(Im(s))/omega,omega
=infinity),symbolic),[r,Cm,xi,p]);
```

calculating differences between the DC-β_{1_2}, β_{1_2}-∞ and DC-∞ plateaus (optional)

reinitializing variables
```
> p:='p': xi:='xi': omega:='omega': gm:='gm': ee:='ee':
ei:='ei':
```
calculation of conductivity and permittivity
```
> sigma_d:=simplify(sigma_inf-sigma_low,symbolic);
> epsilon_d:=simplify(epsilon_inf-epsilon_low,symbolic);
```

simplifications of equations (optional)

factorization of the equations
```
> factor(numer(sigma_d)); factor(denom(sigma_d));
> factor(numer(epsilon_d)); factor(denom(epsilon_d));
```

assigning values to variables, e.g. for a single shell sphere

define physical constants here:
```
> e0:=8.854e-12:
```
radii of the semiaxis
never use more than three positions after decimal point to avoid erroneous results in the calculations of the influential radii
```
> ra:=5.000e-6: rb:=4.999e-6: rc:=5.001e-6:
```

derivations of the relative influential radii of the semi axes of the objects

definitions of the semi axis
```
> a:=ra:
> b:=rb:
> c:=rc:
> Vabc:=4/3*evalf(Pi)*a*b*c:
```

> # definition of the Legendre-Integrals of the 1st and 2nd order for determining the depolarizing coefficients of the general ellipsoid
```
> beta:=(a,b)->b/a: delta:=(a,c)->c/a:
> LF:=(k,theta) -> Int(1/(sqrt(1-
k^2*(sin(psi)^2))),psi=0..theta):
> LE:=(k,theta) -> Int((sqrt(1-
k^2*(sin(psi)^2))),psi=0..theta):
> k:=(beta,delta) -> ((1-beta^2)/(1-delta^2))^(0.5):
> psi:=(delta) -> arccos(delta):
```

> # depolarizing coefficients and influential radii in the direction of the semi axis (a,b,c)
```
> na:=(beta,delta) -> beta*delta/((1-delta^2)^(0.5)*(1-
beta^2))*(LF(k(beta,delta),psi(delta))-
LE(k(beta,delta),psi(delta))):
> nb:=(beta,delta) -> - na(beta,delta)+beta*delta/((1-
delta^2)^(0.5)*(beta^2-
delta^2))*(LE(k(beta,delta),psi(delta)))-delta^2/(beta^2-
delta^2):
> nc:=(beta,delta) -> - beta*delta/((1-delta^2)^(0.5)*(beta^2-
delta^2))*LE(k(beta,delta),psi(delta))+beta^2/(beta^2-
delta^2):
```

```
> ridr:=(n)->1/(1-n): dr:=(r,n)->(ridr(n)-1)*r: xif:=(n)
->ridr(n)-1:

> na:=evalf(na(beta(a,b),delta(a,c)));
nb:=evalf(nb(beta(a,b),delta(a,c)));
nc:=evalf(nc(beta(a,b),delta(a,c))); n_sum:=evalf(na+nb+nc);
> xia:=evalf(xif(na)); xib:=evalf(xif(nb));
xic:=evalf(xif(nc)); xi_sum:=evalf(xia+xib+xic);
```

media parameters

```
> se:=0.1: ee:=80*e0:
> si:=0.50: ei:=50*e0:
> Cm:=1e-2: gm:=0:
```

volume fraction

```
> p:=0.1:
```

plotting conductivity and relative permittivity of the suspension

defining the circular frequency

```
> omega:=2*Pi*10^logf:
```

plotting

```
> plot(Re(s),logf=2..20, labels=[lg(f) (Hz), sigma (S/m)]);
evalf(sigma_low); evalf(sigma_inf);

> plot(Im(s)/omega/e0, logf=2..20, labels=[lg(f) (Hz),
epsilon[r]]);
evalf(epsilon_low/e0); evalf(epsilon_inf/e0);
```

complex plot of the specific impedance of the suspension

```
> with(plots):
p1:=plot([Re(1/s),Im(1/s),logf=0..11], view=[0..11, 0..-5],
labels=[Re(Z[spec]) (Omega), Im(Z[spec]) (Omega)]):
display([p1]);
```

Hints for final simplifications

for further simplifications of the equations Taylor or Pade approximations can be used, alternatively the obtained terms can be reduced manually

final simplifications, e.g. for binomial equations must be done for the individual equations either by hand or using Mathematica®

Simplifications of the media impedances and the Clausius-Mossotti-factor

In the ALE approach, the equivalent circuits for the internal, membrane and external impedances were simplified by reducing their number of R and C components depending on the considered frequency range. These simplifications reduce the complexity of the Clausius-Mossotti-factor and the calculation time (28). The following schemes show examples for such simplifications in different frequency ranges (compare to Fig. 1 *B*).

DC case:



membrane dispersion:



beyond capacitive membrane bridging:



bulk media dispersion:



infinitely high frequencies: